PharmaSUG 2025 - Paper ST-101

Sensitivity Analysis for Overall Survival

Binal Mehta, Mukesh Patel, Merck & Co., Inc., Rahway, NJ, USA

ABSTRACT

Overall survival (OS) refers to the length of time from either the date of diagnosis or the start of treatment for a disease, such as cancer, during which patients diagnosed with the disease remain alive. It is an objective endpoint that directly evaluates the effectiveness of new treatments in prolonging patient life and is often a crucial criterion used in regulatory approvals for new treatments.

Sensitivity analysis for **overall survival** is a statistical method used to assess how various assumptions and parameters such as handling of censoring, missing data, extreme data points, covariates and subgroups, or model components impact the results. The purpose of sensitivity analysis in this context is to gauge the reliability of conclusions drawn from survival data. This helps statisticians and clinicians determine whether their survival estimates remain robust under various conditions.

In this paper, we will describe two widely used methods at our organization: inverse probability of censoring weighting (IPCW) and two-stage analysis, focusing on how they address the handling of censoring, covariates, and subgroups.

INTRODUCTION

This paper will outline details on how IPCW and two-stage analysis are carried out in late-stage clinical trials. It will review the specification and development of core variables, the step-by-step derivation of the key variables required in the analysis dataset, the programming complexity, and the problems and proposed solutions.

SENSITIVITY ANALYSIS

Sensitivity analyses evaluate the robustness of survival estimates by examining how variations in assumptions, data handling, model choices, and other factors influence the results in survival analysis. Sensitivity analysis enhances the validity and credibility of survival analysis findings by identifying potential biases or weaknesses, ensuring that conclusions remain reliable. Below are the key aspects of sensitivity analysis in survival analysis.

1. Handling of Censoring:

Censoring occurs when the event of interest (e.g. death) has not occurred for study participants by the end of the study or the information about the time to event is incomplete, such as when a participant drops out or remains alive. Sensitivity analysis can explore different methods of handling censoring, such as:

- **Different censoring assumptions**: For instance, examining if using a "right-censoring" assumption (the time until an event is either not yet observed or censored) impacts survival estimates.
- **Censoring mechanisms**: Sensitivity analysis can test how different mechanisms of censoring (informative vs. non-informative) might affect the results.

2. Missing Data:

Missing data, such as missing covariates or survival times, is common in survival analysis. Sensitivity analysis can assess how various approaches—such as complete case analysis, imputation, or alternative modeling strategies—impact the results.

3. Choice of Model:

Survival analysis can employ various models, such as the Cox proportional hazards model or parametric survival models like exponential and Weibull. Sensitivity analysis helps evaluate the robustness of findings across different model choices.

- **Proportional hazards assumption**: The Cox model assumes a constant hazard ratio over time. Sensitivity analysis can examine how results change if this assumption is violated, such as by incorporating time-varying covariates or using stratified models.
- **Alternative model fitting**: Comparing Cox model results with those from parametric models (e.g., Weibull, log-normal) helps evaluate the impact of different assumptions about the underlying survival time distribution.

4. Outliers and Influential Data Points:

Extreme data points or outliers, such as unusually long or short survival times, can disproportionately impact survival analysis results. Sensitivity analysis can assess how including or excluding these outliers affects the findings and whether the conclusions remain robust.

5. Subgroup Analysis:

Sensitivity analysis can examine how survival estimates differ across baseline characteristics and subgroups (e.g., age, sex, race, ethnicity, disease stage) to ensure that overall conclusions are not influenced by specific groups or biased by subgroup differences.

6. Assumptions About Covariates:

In general, covariate relationships such as treatment group, age, sex, race, ethnicity with the outcome may be assumed to be linear or more complex. Sensitivity analysis can assess how altering these model assumptions, such as considering linear versus non-linear relationships, impacts survival estimates.

7. Time-varying Effects:

Some survival models assume that a covariate's effect remains constant over time, but this may not always be true. Sensitivity analysis can assess whether allowing time-varying effects alters the results and how dependent the conclusions are on this assumption.

8. Reproducibility and Statistical Variability:

Sensitivity analysis can evaluate the stability of survival estimates across different random samples, data subsets, or bootstrapping methods. This helps determine whether the model is overfitting or if the results are reliable and generalizable.

IPCW (Inverse Probability of Censoring Weighting) METHOD

IPCW is a method used in survival analysis and other types of time-to-event data, particularly when the data is censored. Censoring occurs when the event of interest such as death does not occur for some participants due to various reasons, such as the study ending before the event happens or the participants dropping out.

IPCW helps correct for censoring bias by assigning weights to observations according to the likelihood that an individual was not censored. In essence, it adjusts the analysis to account for censored data, leading to more accurate estimates of treatment effects or survival times.

Key Steps in IPCW:

Calculate the probability of censoring: For each participant, determine the probability that they are observed (i.e., not censored). This can be done using methods like the Kaplan-Meier estimator or the Cox model, depending on the context.

Weight calculation: The inverse of these probabilities is used as weights. Participants with a low probability of being censored receive higher weights in the analysis, while those with a high probability of being censored are assigned lower weights.

Weighted Analysis: Conduct the analysis by applying the weights. This often involves using weighted Cox regression, weighted Kaplan-Meier estimators, or other relevant statistical techniques.

Design of ADIPCW dataset:

We are outlining here derivation details of analysis dataset which includes required variables, and the statistical models used for the IPCW analysis.

Participants level information—such as subject identifier, randomization, death date, date of treatment discontinuation, new anti-cancer therapy information, end of study date and treatment-related data—was obtained from the ADSL (Subject-Level Analysis Dataset) dataset. Additionally, overall survival-related information such as analysis date for overall survival, analysis value, censor, and event description were obtained from the ADTTE (Data for the Time to Event Analyses) dataset and progression of disease date was obtained from the ADINTDT (Intermediate Dates Dataset for ADTTE) dataset.

To provide a clear understanding of how the IPCW method recalculates censor weight for sensitivity analysis, the step-by-step derivations for a single subject 001 are explained.

USUBJID	Unique Subject Identifier
RANDDT	Randomization Date
PDDT	Date of Progression
SWITCHDT	Date Switched to New Anti-Cancer Therapy
CROFL	Switched to New Anti-Cancer Therapy flag
DTHDT	Death Date
ADT	Analysis Date for Overall Survival
AVAL	Analysis value derived as the difference between ADT and Randomization Date in Days
CNSR	Assigned as 0 if event of interest occurred, otherwise assigned as 1

USUBJID	RANDDT	PDDT	SWITCHDT	CROFL	DTHDT	ADT	AVAL	CNSR
001	2019-11-18	2021-02-19	2021-08-19	Y	2021-12-27	2021-12-27	771	0

Display 1. Subject 001 time-to-event data

For this subject 001, overall survival analysis date is assigned based on the date of death, 2021-12-27, however this subject switched to new anti-cancer therapy on 2021-08-19. The overall survival based on IPCW method requires additional variables beyond the core set defined in the CDISC ADAM Basic Data Structure for Time-to-Event Analyses and is calculated as follows. The

calculation requires expanding the data from day 1 to AVAL (for this subject it is 1 to 771) and deriving variables such as ANWK, START, STOP, CNSRI, CYCLE, DEAD, EOSDT, DISCNDT, TTPROG, and PDSTAT. Highlighted orange box represents time to progression and green box represents time to switch to new anti-cancer therapy. Both these records will help in understanding how the censor weight derived

ANWK	Expand the number of observations from 1 to AVAL (Days)
START	From 1 to ANWK
STOP	From 2 to ANWK+1
CNSRI	Assign value of CNSR when ANWK is greater than or equal to AVAL, otherwise assigned value as 1
CYCLE	ANWK/Number of days in Treatment Cycle (21) and round up to larger integer value
DEAD	[1 – CNSRI] from the next ANWK
EOSDT	End of study date
DISCNDT	Date of treatment discontinuation
TTPROG	[(PDDT-RANDDT+1) - ANWK + 1], if PDDT is missing then TTPROG = (EOSDT-RANDDT+1) - ANWK + 1
PDSTAT	Assigned value 'N' if participant did not have PD or did not have PD yet at that timepoint. Assigned value 'Y' from the day of PD onwards.

USUBJID	ANWK	START	STOP	CYCLE	DEAD	TTPROG	PDSTAT	EOSDT	DISCNDT	CNSR	CNSRI
001	1	1	2	1	0	460	N	2021-12-27	.	0	1
001	2	2	3	1	0	459	N	2021-12-27	.	0	1
001	461	461	462	22	0	0	Y	2021-12-27	.	0	1
001	641	641	642	31	0	-180	Y	2021-12-27	.	0	1
001	770	770	771	37	1	-309	Y	2021-12-27	.	0	1
001	771	771	772	37	.	-310	Y	2021-12-27	.	0	0

Display 2. Subject 001 IPCW data.

To derive censor weight, additional intermediate variables must be derived. Some of the key variables are highlighted in display 3 for better understanding.

SWITWK	Difference between (SWITCHDT and RANDDT) +1
CENS	If SWITWK is equal to ANWK then assigned 1, else if ANWK is less than SWITWK then assigned as 0 otherwise keep it missing
PARAM	Overall Survival Adjusted for subsequent treatment
EVNTDESC	Assign as "Not switched" when ANWK is lesser than SWITWK and CNSRI=1, else assigned as "Switched" when ANWK is greater than or equal to SWITWK and CNSRI=1, otherwise assigned value as "Died" when CNSRI=0
PFSCRO	Assigned as 1 when event description for participants is Documented Progression based on independent assessor

CNSRPD	Assigned as 1 when participants are in initial treatment phase, assigned 0 when Participant has PD after switching to new anti-cancer therapy
CROQUAL	Assigned as 1 when participants are in initial treatment phase and PFSCRO, equal to 1 assigned 0 when participants are in initial treatment phase and PFSCRO not equal to 1. Set to missing when participant switched to new anti-cancer therapy

USUBJID	PARAM	ANWK	SWITWK	CENS	PFSCRO	CROFL	CNSRPD	CROQUAL	CNSR	CNSRI	EVNTDESC
001	Overall Survival Adjusted for subsequent treatment	1	641	0	1	Y	1	1	0	1	Not switched
001	Overall Survival Adjusted for subsequent treatment	2	641	0	1	Y	1	1	0	1	Not switched
001	Overall Survival Adjusted for subsequent treatment	461	641	0	1	Y	0	1	0	1	Not switched
001	Overall Survival Adjusted for subsequent treatment	641	641	1	1	Y	0	.	0	1	Switched
001	Overall Survival Adjusted for subsequent treatment	770	641	.	1	Y	0	.	0	1	Switched
001	Overall Survival Adjusted for subsequent treatment	771	641	.	1	Y	0	.	0	0	Died

Display 3. Subject 001 IPCW data

CNSRWGT (Weights Product Limit) derivation explanation:

The steps for derivation of CNSRWGT are below. The variables AGE, RACE, REGION and ECOG are from the ADSL dataset and not shown in Display 3.

The value of CENS at time t is to be regressed against the covariates at t-1.

Calculate CNSRWGT = PL (numerator) / PL (denominator). Here PL () is by subject and indicates a product-limit i.e. the probability for next time point is the product of probability at that time and the one before. The probability for first time point is unchanged.

For the numerator calculation, run pooled logistic regression for CENS(t) = AGE RACE REGION ANWK(t-1) and predict probability of CENS(t)=0.

For the denominator calculation, run pooled logistic regression for CENS(t) = AGE RACE REGION CNSRPD ECOG (t-1) ANWK(t-1), and predict probability of CENS(t)=0.

CNSRWGT should then be accumulated as product overtime as below:

Numerator Calculation:

```
proc logistic data=final
(keep= usubjid age race region cnsrpd ecog censt anwk
 where=(anwk > 0)) noprint;
class age race region /param=ref ref=first;
model censt(event='0') = age race region anwk /firth;
output out=final_n(keep=usubjid anwk pl_num) pred=pl_num;
run;
```

Denominator Calculation:

```
proc logistic data=final
(keep= usubjid age race region cnsrpd ecog censt anwk
 where=(anwk > 0)) noprint;
       class age race region /param=ref ref=first;
       model censt (event='0') = age race region cnsrpd ecog anwk
      /firth maxiter=150;
       output out=final_d(keep=usubjid anwk pl_denom)
pred=pl_denom;
run;

proc sort data = final_n;  by usubjid anwk; run;

proc sort data = final_d;  by usubjid anwk; run;

data final_1;
     merge final_n final_d;
     by usubjid anwk;
     if nmiss(pl_denom,pl_num)=0 then cwgt = pl_num/pl_denom;
        cnsrwgt_pre=lag(cwgt);
        if first.usubjid then cnsrwgt_pre=.;
run;

proc sort data = final;  by usubjid anwk; run;

data main;
     merge final final_1(keep = usubjid anwk cnsrwgt_pre);
     by usubjid anwk;
         /* finalise cnsrweight */
        fcnsrwgt=cnsrwgt_pre;
        if fcnsrwgt=. then fcnsrwgt=1;

        %* accumulated weight;
        retain aclwgt;
        if first.usubjid then aclwgt=fcnsrwgt;
        else aclwgt=aclwgt*fcnsrwgt;

        cnsrwgt=aclwgt;
        if anwk >= switwk then cnsrwgt=0;
run;
```

A*B = C

USUBJID	PARAM	SWITWK	ANWK	CENS	CNSRWGT_PRE	FCNSRWGT	ACLWGT	CNSRWGT
001	Overall Survival Adjusted for subsequent treatment	641	1	0	.	1	1	1
001	Overall Survival Adjusted for subsequent treatment	641	2	0	0.9996953278	0.9996953278	0.9996953278 (B)	0.9996953278
001	Overall Survival Adjusted for subsequent treatment	641	3	0	0.999695036	0.999695036 (A)	0.9993904567 (C)	0.9993904567
001	Overall Survival Adjusted for subsequent treatment	641	4	0	0.9996947451	0.9996947451	0.9990853878	0.9990853878
001	Overall Survival Adjusted for subsequent treatment	641	461	0	1.0013968385	1.0013968385	0.8545939517	0.8545939517
001	Overall Survival Adjusted for subsequent treatment	641	641	1	1.0009773916	1.0009773916	1.0558321712	0
001	Overall Survival Adjusted for subsequent treatment	641	770	.	1.0007459592	1.0007459592	1.1791320345	0
001	Overall Survival Adjusted for subsequent treatment	641	771	.	1.0007443559	1.0007443559	1.1800097284	0

Display 4. Subject 001 Censor Weight computations

Overall Survival Time: Censor weight calculated in above step and highlighted in display 4 is used in calculation of overall survival time by accumulation as describe below.

```
proc means data = main;
var cnsrwgt;
by usubjid param;
output out = totcnsrwgt(keep = usubjid param ovrlos_) sum=ovrlos_;
run;

data adipcw;
     merge main totcnsrwgt;
by usubjid param;

   if switchdt ^= . and anwk = switwk then evntfln=1;
      else if switchdt  ^= . and switwk> anwk and aval=anwk
           then  evntfln=1;
      else if switchdt = . and anwk = aval then evntfln=1;

   if evntfln=1 then ovrlos=ovrlos_;

run;
```

USUBJID	PARAM	SWITWK	ANWK	AVAL	EVNTFLN	CNSRWGT	OVRLOS
001	Overall Survival Adjusted for subsequent treatment	641	641	771	1	0	602.72943797

Display 5. Subject 001 recalculation of OS analysis value using IPCW method

For the subject 001, Overall survival days change from 771 to 602.73 as highlighted in display 5.

Design of AD2STG (Two-Stage Method) dataset

This section provides the derivation details of the AD2STG analysis dataset, highlighting the key variables needed for the two-stage analysis and the statistical models applied.

Participants level information such as subject identifier and randomization information are needed from the ADSL (Subject-Level Analysis Dataset) dataset and overall survival related information such as analysis date for overall survival and censor, and progression of disease date are needed from the ADINTDT (Intermediate Dates Dataset for ADTTE) dataset. Further derived variables such as AVALPD, CROFLN and TTPROG as describe below are required to feed into the statistical model along with the baseline variables such as age group (AGEGR), race group (RACEGR), region (REGION), baseline ECOG value (ECOGBL) from the ADSL dataset.

The Weibull and Lognormal statistical model are used and derive acceleration factor (ACFA). Two parameters (1) Overall Survival Adjusted for subsequent treatment and (2) Overall Survival Adjusted for subsequent treatment – Recensored with the help of acceleration factor and recalculated analysis value and censoring. These two parameters have been used in Kaplan-Meier method to re-calculate overall survival analysis.

For better understanding, derivations for subjects 002 and 003 are explained in step-by-step detail.

USUBJID	Unique Subject Identifier
RANDDT	Randomization Date
PDDT	Date of Progression
TTPROG	Analysis value derived by difference between Date of Progression and Randomization Date
ADT	Date for Overall Survival
AVALPD	Analysis value derived by difference between ADT and Date of Progression
CNSR	Assigned as 0 if event of interest occurred, otherwise assigned as 1

USUBJID	RANDDT	PDDT	ADT	AVALPD	TTPROG	CNSR
002	2020-05-26	2021-01-14	2022-09-20	615	234	0

USUBJID	RANDDT	PDDT	ADT	AVALPD	TTPROG	CNSR
003	2020-06-22	2021-05-08	2023-08-23	838	321	0

Display 6. Subject 002 and 003 data for Two-stage Method

Additional variables are derived as described below to feed into the statistical models.

AVALO	Analysis value derived by difference between ADT and Randomization Date
PARAM	Overall Survival Adjusted for subsequent treatment
CROFL	Switched to New Anti-Cancer Therapy flag
CROFLN	Switched to New Anti-Cancer Therapy flag (N)
SWITCHDT	Switched Date to New Anti-Cancer Therapy

USUBJID	PARAM	CROFL	CROFLN	RANDDT	PDDT	SWITCHDT	AVALO
002	Overall Survival Adjusted for subsequent treatment	Y	1	2020-05-26	2021-01-14	2021-03-11	848

USUBJID	PARAM	CROFL	CROFLN	RANDDT	PDDT	SWITCHDT	AVALO
003	Overall Survival Adjusted for subsequent treatment	Y	1	2020-06-22	2021-05-08	2021-10-12	1158

Display 7. Subject 002 and 003 additional variables for Two-stage method

```
ods listing close;
ods output  parameterestimates=__weibull_crofln.
FitStatistics=__wbff_crofln;

proc lifereg data = main;
  where pddt ne '';
  class agegr racegr region ecogbl crofln;
  model avalpd * cnsr(1) = crofln agegr racegr region ecogbl ttprog/
dist=weibull;
run;

ods listing close;
ods output  parameterestimates=__lognorm_crofln
FitStatistics=__lnff_crofln;

proc lifereg data = main;
 where pddt ne '';
 class agegr racegr region ecogbl crofln;
 model avalpd * cnsr(1) = crofln. agegr racegr region ecogbl ttprog/
dist= lognormal;
run;

ods output close;
ods listing;

data __best_fit;
 merge __lnff_ crofln(rename=(value=lognormal)) __wbff_
crofln(rename=(value=weibull));
  by criterion;

  if criterion = 'AIC (smaller is better)';

  if weibull <= lognormal then call symputx('bestfit','WEIBULL');
  else if weibull > lognormal then call
symputx('bestfit','LOGNORM');
run;

data main2;
 set __&bestfit._crofln;
      where parameter = upcase("crofln") and level1 = '1';

      acfa = exp(estimate);
      acfalcl = exp(lowercl);
      acfaucl = exp(uppercl);
run;
```

In the code above, the models calculate parameter estimates (*parameterestimates*=) and fit statistics (*FitStatistics*=) for both the Lognormal and Weibull methods. Next, a comparison is made of the fit statistics from both models, selecting the smaller value between them for further analysis as shown in below.

CRITERION	LOGNORMAL	WEIBULL
AIC (smaller is better)	553.708	582.478

Display 8. AIC values for both the models

In Display 8, the Lognormal method is chosen as it has lower fit statistics value. Using parameter estimates from Lognormal method, the variables *ACFA, ACFALCL and ACFAUCL* are derived, and values are shown in Display 9.

PARAMETER	LEVEL1	DF	ESTIMATE	STDERR	LOWERCL	UPPERCL	CHISQ	PROBCHISQ	ACFA	ACFALCL	ACFAUCL
CROFL1N	1	1	0.2108	0.2367	-0.2531	0.6747	0.79	0.3731	1.2346973925	0.7764249079	1.963457941

Display 9. Acceleration factor values for lognormal method

Next a recalculation of analysis value (AVAL) for parameter "Overall Survival Adjusted for subsequent" treatment is shown below and results for Subjects 002 and 003 are highlighted in Display 10.

```
if param in ('Overall Survival Adjusted for
subsequent treatment' ) then do;
   if crofl1 = 'Y' and
nmiss(adt,crodt1,acfa,randdt)=0 then
        aval = (adt - crodt1)*(1/acfa)+(crodt1-
randdt)+1;
```

USUBJID	PARAM	RANDDT	SWITCHDT	ADT	CROFL	CROFLN	AVALO	AVAL	ACFA	CNSR
002	Overall Survival Adjusted for subsequent treatment	2020-05-26	2021-03-11	2022-09-20	Y	1	848	741.93259772	1.2346973925	0

USUBJID	PARAM	RANDDT	SWITCHDT	ADT	CROFL	CROFLN	AVALO	AVAL	ACFA	CNSR
003	Overall Survival Adjusted for subsequent treatment	2020-06-22	2021-10-12	2023-08-23	Y	1	1158	1028.7422338	1.2346973925	0

Display 10. Subject 002 and 003 recalculation of analysis value

Derivation of parameter "Overall Survival Adjusted for subsequent treatment – Recensored"

Now using the value of the acceleration factor (ACFA) and analysis value of parameter "Overall Survival Adjusted for subsequent treatment", we are deriving new parameter "Overall Survival Adjusted for subsequent treatment – Recensored" and adjusting the censor (CNSR), event description (EVNTDESC) and analysis value (AVAL)

```
if acfa >= 1 then do;
            if nmiss(cutoffdt,randdt)=0 then c0 = (cutoffdt - randdt
      +1)*(1/acfa);
            if nmiss(avalo,randdt)=0 then c1 = (avalo)*(1/acfa);

            if cnsr=0 and aval > c0 then do;
              aval=c0;
              cnsr=1;
              evntdesc = 'Censored';
            end;
            else if cnsr=1 and aval > c1 then do;
              aval=c1;
              cnsr=1;
              evntdesc = 'Censored';
            end;
            else do;
              aval=aval;
            end;
          end;
          else if acfa < 1 then do;
            if nmiss(cutoffdt,randdt)=0 then c0_ = cutoffdt - randdt +1;
            if nmiss(avalo,randdt)=0 then c1_ = (avalo);

            if cnsr=0 and aval > c0_ then do;
              aval=c0_;
              cnsr=1;
              evntdesc = 'Censored';
            end;
            else if cnsr=1 and aval > c1_ then do;
              aval=c1_;
              cnsr=1;
              evntdesc = 'Censored';
            end;
          end;
          else do;
            aval=aval;
          end;
        end;
```

Above is the calculation for the new re-censoring parameter derivation. For subject 002, analysis value (AVAL) and censor (CNSR) remain same but for subject 003 it changes as highlighted in Display 11.

USUBJID	PARAM	RANDDT	ADT	CUTOFFDT	DTHDT	CNSR	AVALO	ACFA	EVNTDESC	AVAL	C0	C0_	C1	C1_
002	Overall Survival Adjusted for subsequent treatment - Recensored	2020-05-26	2022-09-20	2023-10-02	2022-09-20	0	848	1.2346973925	Death	741.93259772	992.14593585	.	686.80796212	.

USUBJID	PARAM	RANDDT	ADT	CUTOFFDT	DTHDT	CNSR	AVALO	ACFA	EVNTDESC	AVAL	C0	C0_	C1	C1_
003	Overall Survival Adjusted for subsequent treatment - Recensored	2020-06-22	2023-08-23	2023-10-02	2023-08-23	1	1158	1.2346...	Censored	970.278...	970.278...	.	937.881...	.

Display 11. Subject 002 and 003 recalculation of analysis value for re-censoring parameter

CHALLENGE AND RESOLUTION:

Size of the data: While working on ADaM datasets, statistical models were run multiple times until they were finalized. Since these models are time-consuming to execute, we recommend running them on a smaller sample size until the final version is determined.

Notes and Warnings from Model: While working with these models, various warnings were encountered, such as 'Convergence was not attained in xx iterations.' Our recommendation is to carefully review these warnings and consult with a statistician to determine the appropriate resolution or acceptance.

Frequent Updates: Since this type of analysis is uncommon, statisticians regularly updated the specifications to identify the best-fit model. It is important to keep track of all changes to ensure the ability to revert to previous versions if necessary.

Timelines and Execution: Given the complexity of the analysis and the time required to run these statistical models, planning in advance is recommended and securing resources for both development and validation.

Collaboration: Due to the complexity of the analysis, it is essential to work closely with statisticians and the clinical team, ensuring clear communication regarding updates to specifications and statistical models to ensure timely and high-quality deliverables.

CONCLUSION

This paper outlines the methodologies and considerations involved in conducting sensitivity analysis for overall survival (OS) using two widely adopted statistical approaches: Inverse Probability of Censoring Weighting (IPCW) and Two-Stage Analysis. We discussed how both methods effectively address key challenges such as censoring, to ensure robust survival estimates.

IPCW adjusts for censoring bias by applying weights based on the likelihood of being censored, while the Two-Stage method uses statistical models like Weibull and Lognormal to adjust for subsequent treatments. By comparing different models, statisticians can select the best-fitting approach to accurately assess treatment efficacy in clinical trials.

Additionally, we also highlighted challenges, including data size constraints, model warnings, and the need for frequent updates to specifications. Effective planning, collaboration with statisticians and clinical teams, and clear communication are essential for managing the complexity of these analyses and ensuring timely high-quality deliverables.

In conclusion, sensitivity analysis is a critical tool for ensuring the reliability of survival estimates, and the methods discussed provide a robust framework for handling complex survival data. By carefully considering assumptions, choosing appropriate models, and maintaining thorough documentation, statisticians can ensure that conclusions drawn from survival analyses are both valid and trustworthy, supporting the development of effective treatments in clinical trials.

ACKNOWLEDGEMENT:

The authors would like to thank management team Lisa Pyle, Sapan Patel, and Amy Gillespie for their time in reviewing the paper and providing valuable comments.

REFERENCES

National Library of Medicine
https://pmc.ncbi.nlm.nih.gov

Role of Sensitivity Analyses in Assessing Progression-Free Survival in Late-Stage Oncology Trials
https://ascopubs.org/doi/full/10.1200/JCO.2009.22.4329

Inverse Probability of Censoring Weighting for Selective Crossover in Oncology Clinical Trials.
https://www.lexjansen.com/phuse/2014/sp/SP02.pdf

CONTACT INFORMATION

Name: Binal Mehta
Enterprise: Merck & Co., Inc.
Address: 126 E. Lincoln Ave
City, State & Zip: Rahway, NJ 07065-4607 USA
Phone: (732) 594-3383
E-mail: binal.mehta@merck.com
Web: www.merck.com

Name: Mukesh Patel
Enterprise: Merck & Co., Inc.
Address: 126 E. Lincoln Ave
City, State & Zip: Rahway, NJ 07065-4607 USA
Phone: (732) 594-3057
E-mail: mukesh.patel1@merck.com
Web: www.merck.com